
第七章  字串函數 
 
1. 字串相加 
 
【例題 7-1】 

 
LIST 
1  '************* string test ********* IBM.BAS 
10 A$ = "IBM " 
20 B$ = "Personal " 
30 C$ = "Computer" 
40 PRINT A$ + B$ + C$ 
Ok 
RUN 
IBM Personal Computer 
Ok 

 
 
2. 字串長度 
 

LEN(字串)  
 
【例題 7-2】 

 
PRINT LEN("123") 
 3 
Ok 
A$ = "Hello" 
Ok 
PRINT LEN (A$) 
 5 
Ok 

 
 
3.將數值轉換成 ASC II 碼 
 

CHR$(n) {將 n 值轉換成 ASC II 碼}       n 值範圍在 1 - 255 

【例題 7-3】 
 

PRINT CHR$(65) 
A 
Ok 
PRINT CHR$(66) 
B 
Ok 

 

 

【例題 7-4】利用CHR$( ) 印出 A - Z  
 

LIST 
10 '***************CHR.BAS 
20 FOR I = 65 TO 90 
25     IF I MOD 5 = 0 THEN PRINT 
30     PRINT USING "###:!   "; I; CHR$(I); 
40 NEXT I 
50 END 
Ok 
 

RUN 
 
 65:A    66:B    67:C    68:D    69:E 
 70:F    71:G    72:H     73:I     74:J 
 75:K    76:L    77:M    78:N    79:O 
 80:P    81:Q    82:R     83:S     84:T 
 85:U    86:V    87:W    88:X    89:Y 
 90:Z 
Ok 

 
 
4.顯示 ASCII 的值 
 

ASC(字元) {顯示 ASCII 的值} 
 
【例題 7-5】 

 
LIST 
10 '*** asc test *** asc-1.bas 
20 CLS 
30 INPUT "INPUT A STRING : ", A$ 
40 IF A$ = "" THEN 99 
              ELSE PRINT ASC(A$) 
50 GOTO 30 
99 END 
Ok 

RUN 
INPUT A STRING : 0 
 48 
INPUT A STRING : A 
 65 
INPUT A STRING : a 
 97 
INPUT A STRING : 
Ok 

5.轉為數值 
 

VAL(字串)  {將字串轉為數值} 
 
【例題 7-6】 

 
LIST 
10 A=123 
20 B$="123" 
30 PRINT A 
40 PRINT B$ 
50 PRINT VAL(B$) 
Ok 

RUN 
 123 
123 
 123 
Ok 

 

 

【例題 7-7】 
 

LIST 
10 '*** val test *** val-1.bas 
20 CLS 
30 LOCATE 7,1 
40 PRINT TAB(22); "HEXADECIMAL"; TAB(40); "DECIMAL" 
50 Y = 8: X = 25 
60 LOCATE Y, X: INPUT A$ 
70 IF A$ = "/" THEN 120 
80 H = VAL("&H" + A$) 
90 LOCATE Y,X+15: PRINT H 
100 Y = Y + 1 
110 GOTO 60 
120 LOCATE Y+2, 30 
130 PRINT "*** END ***" 
140 END 
Ok 
RUN 
HEXADECIMAL       DECIMAL 
   ? 10                           16 
   ? A                            10 
   ? FF                          255 
   ? 123                        291 
   ? / 
        *** END *** 

6.化為字串 
 

STR$(n) {將數值轉為字串} 
 
【例題 7-8】 

 
10 A=12 : B = 34 
20 PRINT A + B 
30 PRINT STR$(A) + STR$(B) 
RUN 
 46 
 12 34 
Ok 

 
 
7.重覆字元 
 

STRING$(n, 字元)  {重覆 n 個字元} 
         n 值範圍在 1 - 255 
        字元 --> 可為字元或字元變數 

 
  【例題 7-9】 

 
PRINT STRING$(12,"A") 
AAAAAAAAAAAA 
Ok 
PRINT STRING$(20,"=") 
==================== 
Ok 

  
 
8.重覆空白字元 
 

SPACE$(n)  {重覆 n 個空白}       n (1 - 255) 
 
【例題 7-10】 

 
PRINT "          "; "Hello !" 
          Hello ! 
Ok 
PRINT SPACE$(10); "Hello !" 
          Hello ! 

【例題 7-11】SPACE$( ) 的應用  
 

LIST 
10 FOR I = 1 TO 5 
20     A$ = SPACE$(I) 
30     PRINT A$; I 
40 NEXT I 
99 END 
Ok 

RUN 
  1 
   2 
    3 
     4 
      5 
Ok 

 
 
【例題 7-12】STRING$( ) 與 SPACE$( ) 的應用 (一)  
 

LIST 
 10 FOR I = 1 TO 10 
 20     PRINT SPACE$((10 - I)); 
STRING$(I, "*") 
 30 NEXT I 
 Ok 

RUN 
          * 
         ** 
        *** 
       **** 
      ***** 
     ****** 
    ******* 
   ******** 
  ********* 
 ********** 
 Ok 

  
  
【例題 7-13】STRING$( ) 與 SPACE$( ) 的應用 (二)  
 

LIST 
10 FOR I = 1 TO 10 
20     PRINT SPACE$((10 - I)); 
STRING$((I*2)-1,"*") 
30 NEXT I 
Ok 

RUN 
         * 
        *** 
       ***** 
      ******* 
     ********* 
    *********** 
   ************* 
  *************** 
 ***************** 
******************* 
Ok 

9.尋找字串位置 
 

INSTR(n, 字串1, 字串2)  {尋找字串位置} 
 
* 從字串1中, 第 n 個字元開始尋找與字串 2 相同的字串位置, 若無則傳回 0. 
 
【例題 7-14】 

 
LIST 
10 A$ = "I Love Personal Computer" 
20 B$ = "Love" 
30 N = INSTR(A$, B$) 
40 PRINT N 
Ok 
RUN 
 3 
Ok 

 
 
【例題 7-15】 

 
LIST 
10 '*** INSTR test *** instr-1.bas 
12 CLS 
15 PRINT "DO YOU LOVE ME !!! (Y/N) ? " 
20 A$ = INPUT$(1) 
30 IF INSTR("YyNn", A$) = 0 THEN BEEP: 
                            PRINT "TRY AGAIN !!!": 
                            GOTO 15 
40 IF INSTR("Nn", A$) THEN 100 
50 PRINT "I LOVE YOU, TOO !" 
60 GOTO 110 
100 PRINT "YOU A PIG !" 
110 END 
Ok 

RUN 
DO YOU LOVE ME !!! (Y/N) ? 
TRY AGAIN !!! 
DO YOU LOVE ME !!! (Y/N) ? 
I LOVE YOU, TOO ! 
Ok 

10.取左邊字串 
 

LEFT$(字串,n)  {取左邊字串}     *從字串左邊, 取 n 個字元 
 
【例題 7-16】 

 
LIST 
10 A$ = "ABCDEFG" 
20 FOR I = 1 TO LEN(A$) 
30     PRINT LEFT$(A$, I) 
40 NEXT I 
50 END 
Ok 

RUN 
A 
AB 
ABC 
ABCD 
ABCDE 
ABCDEF 
ABCDEFG 
Ok 

 
 
11.取右邊字串 
 

RIGHT$(字串,n)  {取右邊字串}     *從字串右邊, 取 n 個字元 
 
【例題 7-17】 

 
LIST 
10 A$ = "1234567" 
20 FOR I = LEN(A$) TO 1 STEP -1 
30     PRINT RIGHT$(A$, I) 
40 NEXT I 
50 END 
Ok 

RUN 
1234567 
234567 
34567 
4567 
567 
67 
7 
Ok 

12. 取中間字串 
 

MID$(字串,n,m)  {取中間字串}   *從字串左邊第 n 個字元, 取 m 個字元 
 
【例題 7-18】 

 
LIST 
10 INPUT "INPUT A STRING : ", A$ 
20 FOR I = LEN(A$) TO 1 STEP -1 
30     B$ = B$ + MID$(A$,I,1) 
40 NEXT I 
50 PRINT B$ 
Ok 

RUN 
INPUT A STRING : ABCDEFG 
GFEDCBA 
Ok 
RUN 
INPUT A STRING : 12345 
54321 
Ok 

 
 
【例題 7-19】活動廣告 (一)  
 

LIST 
1  '**************** DEMO1.BAS 
10  CLS 
20  DEMO$ = "IBM COMPUTER BASICA" 
30  X = 1: Y = 12: F = 0: F1 = 0 
40  LOCATE Y-2, 1: PRINT STRING$(79,"*") 
50  LOCATE Y+2, 1: PRINT STRING$(79,"*") 
60  LOCATE Y,X : PRINT DEMO$ 
70  FOR I = 1 TO 50: NEXT I 
80  LOCATE Y,X : PRINT "                   " 
90  IF X > 60 THEN F = 1 
100 IF X <=  1 THEN F = 0 
110 IF F THEN X = X - 2 
         ELSE X = X + 2 
120 GOTO 60 
999 END 
Ok 
RUN 
************************************************* 
 
  IBM COMPUTER BASICA 
 
************************************************* 

【例題 7-20】活動廣告 (二)  
 

LIST 
10  ' SUPER COMPUTER 
20  A$ = "SUPER COMPUTER ..." 
30  A = LEN(A$) 
40  Z$ = STRING$(A,"=") 
50  S = 300 
60  CLS: KEY OFF 
70  LOCATE 10,20: PRINT STRING$(A, "=") 
80  LOCATE 12,20: PRINT Z$ 
90  FOR I = 1 TO A 
100     D$ = MID$(A$, I, A) + MID$(A$, 1, I-1) 
110     LOCATE 11, 20: PRINT D$ 
120     FOR J = 1 TO S: NEXT J 
130     ZZ$ = INKEY$ 
140     IF ZZ$ = CHR$(13) THEN 900 
300 NEXT I 
310 GOTO 90 
900 KEY ON 
999 END 
Ok 

  RUN 
  ================== 
  SUPER COMPUTER ... 
  ================== 
 
 

  ================== 
  COMPUTER ...SUPER 
  ================== 

 


