
第六章  陣列 
 
1. DIM  變數(n,m,..)  {宣告陣列}        變數 ==> 可為數值或字串變數 

 
* DIM  變數(n)       ==> 一維陣列 
* DIM  變數(n, m)  ==> 二維陣列 
 
【例題 6-1】 
 

LIST 
1  '********** DIM-1 *********** DIM-1.BAS 
10 DIM S(5) 
20 PRINT "******* INPUT DATA ******" 
30 FOR I = 1 TO 5 
40     PRINT I; ": "; 
50     INPUT "",S(I) 
60 NEXT I 
70 PRINT "******* PRINT DATA ******" 
80 FOR I = 1 TO 5 
90     PRINT I; ":"; S(I) 
100 NEXT I 
Ok 
 

RUN 
******* INPUT DATA ****** 
 1 : 90 
 2 : 80 
 3 : 88 
 4 : 98 
 5 : 85 
******* PRINT DATA ****** 
 1 : 90 
 2 : 80 
 3 : 88 
 4 : 98 
 5 : 85 
Ok 

【例題 6-2】輸入 10 個數值, 判斷奇偶數 
 

LIST 
10  '************** ODD & EVEN *********** ODD.BAS 
20  CLS 
30  DIM A(10), B(10) 
40  FOR I = 1 TO 10 
50      READ X 
60      Q = X / 2 
70      IF Q = INT(Q) THEN 110 
80      N = N + 1 
90      A(N) = X 
100     GOTO 130 
110     M = M + 1 
120     B(M) = X 
130 NEXT I 
140 ' 
150 PRINT "ODD : " 
160 FOR I = 1 TO N 
170     PRINT A(I), 
180 NEXT I 
190 ' 
200 PRINT: PRINT "EVEN : " 
210 FOR I = 1 TO M 
220     PRINT B(I), 
230 NEXT I 
300 DATA 22,45,68,24,13 
310 DATA 12,45,20,33,17 
Ok 
ODD : 
 45            13            45            33            17 
 

EVEN : 
 22            68            24            12            20 
Ok 

【例題 6-3】PASCAL 三角形 
 

LIST 
10 ' *** PASCAL test *** pas.bas 
20 CLS 
30 INPUT "KEY IN A NUMBER : ", N 
40 DIM A(N), B(N) 
50 FOR I = 1 TO N 
60     PRINT SPC(N * 2 - I * 2); 
70     FOR J = 1 TO I 
80         IF J = 1 OR J = I THEN A(J) = 1 
                             ELSE A(J) = B(J-1) + B(J) 
90         PRINT USING "####"; A(J); 
100    NEXT J 
110    PRINT 
120    FOR K = 1 TO I 
130        B(K) = A(K) 
140    NEXT K 
150 NEXT I 
160 END 
Ok 

RUN 
KEY IN A NUMBER : 8 
                 1 
               1   1 
             1   2   1 
           1   3   3   1 
         1   4   6   4   1 
       1   5  10  10   5   1 
     1   6  15  20  15   6   1 
   1   7  21  35  35  21  7   1 
Ok 

【例題 6-4】輸入 10 個學生分數, 將及格和不及格分別列出  
 

LIST 
10 '*********** STUDENT SCORE -1 ********** STU-1.BAT 
20  CLS 
30  DIM A(10), B(10) 
40  FOR I = 1 TO 10 
50      READ S 
60      IF S < 60 THEN 100 
70      N = N + 1 
80      A(N) = S 
90      GOTO 120 
100     M = M + 1 
110     B(M) = S 
120 NEXT I 
130 ' 
140 PRINT "PASS: " 
150 FOR I = 1 TO N 
160     PRINT A(I), 
170 T1 = T1 + S 
180 NEXT I 
190 PRINT: PRINT "TOTAL = "; T1 
200 ' 
210 PRINT: PRINT "FAIL: " 
220 FOR I = 1 TO M 
230     PRINT B(I), 
240 T2 = T2 + S 
250 NEXT I 
260 PRINT: PRINT "TOTAL = "; T2 
300 DATA 90,76,56,87,45 
310 DATA 55,78,98,67,60 
Ok 
RUN 
PASS: 
 90            76            87            78            98 
 67            60 
TOTAL =  420 
 
FAIL: 
56            45            55 
TOTAL =  180 
Ok 

【例題 6-5】 
 

LIST 
10  '********** DIM-2 *********** DIM-
2.BAS 
20 CLS 
30 INPUT "ENTER STUDENT NUMBER : ", N 
40 DIM NO$(N), NAM$(N), SCORE(N) 
50 PRINT "******* INPUT DATA ******" 
60 FOR I = 1 TO N 
70     PRINT TAB(10); I 
80     PRINT STRING$(20, "=") 
90     INPUT "NO.   : ", NO$(I) 
100    INPUT "NAME  : ", NAM$(I) 
110    INPUT "SCORE : ", SCORE(I) 
120 NEXT I 
130 PRINT : PRINT 
140 PRINT "******* PRINT DATA ******" 
150 PRINT       " NO.       NAME      SCORE" 
160 PRINT       " ===       =====     =====" 
170 FOR I = 1 TO N 
180     PRINT USING " \ \       \   \     ###.#"; 
                     NO$(I), NAM$(I), SCORE(I) 
190 NEXT I 
200 PRINT       " ===       =====     =====" 
Ok 

RUN 
ENTER STUDENT NUMBER : 3 
******* INPUT DATA ****** 
          1 
==================== 
NO.   : 001 
NAME  : John 
SCORE : 78 
          2 
==================== 
NO.   : 002 
NAME  : Mary 
SCORE : 90 
          3 
==================== 
NO.   : 003 
NAME  : Bob 
SCORE : 80 
 
******* PRINT DATA ****** 
 NO.       NAME    SCORE 
 ===       =====     ===== 
 001       John          78.0 
 002       Mary         90.0 
 003       Bob           80.0 
 ===       =====     ===== 
Ok 

【例題 6-6】 
 

LIST 
10  '********** DIM-3 ********** DIM-3.BAS 
20 CLS 
30 INPUT "ENTER STUDENT NUMBER : ", N 
40 DIM NO$(N), NAM$(N), CHI(N), CHE(N), MAT(N) 
50 PRINT "******* INPUT DATA ******" 
60 FOR I = 1 TO N 
70     PRINT TAB(10); I 
80     PRINT STRING$(20, "=") 
90     INPUT "NO. : ", NO$(I) 
100    INPUT "NAME: ", NAM$(I) 
110    INPUT "CHI : ", CHI(I) 
120    INPUT "CHE : ", CHE(I) 
130    INPUT "MAT : ", MAT(I) 
140 NEXT I 
150 PRINT : PRINT 
155 ' 
160 PRINT "******* PRINT DATA ******" 
170 PRINT " NO.   NAME    CHI    CHE    MAT    AVE  DOWN" 
180 PRINT " ===   =====  =====  =====  =====  ===== ====" 
190 FOR I = 1 TO N 
200     AVE = (CHI(I) + CHE(I) + MAT(I)) / 3 
210     DOWN$ = " " 
220     IF AVE < 60 THEN DOWN$ = "*" 
230     PRINT USING " \ \   \   \  ###.#  ###.#  ###.#  ###.#   !"; 
NO$(I),NAM$(I),CHI(I),CHE(I),MAT(I),AVE,DOWN$ 
240 NEXT I 
250 PRINT  " ===   =====  =====  =====  =====  ===== ====" 
Ok 
RUN 
ENTER STUDENT NUMBER : 3 
******* INPUT DATA ****** 
          1 
==================== 
NO. : 001 
NAME: John 
CHI   : 80 
CHE  : 88 
MAT : 90 
          2 
==================== 
NO. : 002 
NAME: Mary 
CHI   : 80 
CHE  : 60 
MAT : 56 
          3 
==================== 
NO. : 003 
NAME: Bob 
CHI   : 70 
CHE  : 40 
MAT : 50 
******* PRINT DATA ****** 
 NO.   NAME   CHI       CHE     MAT    AVE   DOWN 
 ===   =====  =====  =====  =====  ===== ==== 
 001   John       80.0      88.0       90.0      86.0 
 002   Mary      80.0      60.0       56.0      65.3 
 003   Bob        70.0      40.0       50.0      53.3      * 
 ===   =====  =====  =====  =====  ===== ==== 
Ok 

2.設定陣列的最小註標值 
 

OPTION BASE n  {設定陣列的最小註標值}     n 只可為 0 或 1, 內定為 0 
 

3.清除列陣 
 

ERASE 陣列名稱 [,陣列名稱]  {清除列陣} 
 
【例題 6-7】 
 

LIST 
10 '**** erase test **** reas-1.bas 
20 DIM A(3) 
30 FOR I = 1 TO 3 
40     A(I) = I 
50 PRINT I, 
60 NEXT I 
70 PRINT 
80 DIM A(3) 
90 FOR I = 1 TO 3 
100     A(I) = I 
110 PRINT I, 
120 NEXT I 
Ok 
RUN 
 1             2             3 
Duplicate Definition in 80 
Ok 

【例題 6-8】 
 

LIST 
10 '**** erase test **** reas-1.bas 
20 DIM A(3) 
30 FOR I = 1 TO 3 
40     A(I) = I 
50 PRINT I, 
60 NEXT I 
70 PRINT 
75 ERASE A 
80 DIM A(3) 
90 FOR I = 1 TO 3 
100     A(I) = I 
110 PRINT I, 
120 NEXT I 
Ok 
RUN 
 1             2             3 
 1             2             3 
Ok 

 
 
4.資料互換 
 

 SWAP 變數1, 變數2  {資料互換} 
 
【例題 6-9】 
 

LIST 
10  '***** G.C.D  &  L.C.M *****GCD-1.BAS 
20  INPUT "INPUT A,B : ", A, B 
30  IF A < B THEN T = A:A = B:B = T 
40  BIG = A: SMALL = B 
50  R = BIG MOD SMALL 
60  IF R = 0 THEN 90 
70  BIG = SMALL : SMALL = R 
80  GOTO 50 
90  GCD = SMALL 
100 LCM = A * B / GCD 
110 PRINT "G.C.D. = "; GCD 
120 PRINT "L.C.M. = "; LCM 
Ok 

RUN 
INPUT A,B : 12,8 
G.C.D. =  4 
L.C.M. =  24 
Ok 
RUN 
INPUT A,B : 9,36 
G.C.D. =  9 
L.C.M. =  36 
Ok 

【例題 6-10】 
 

LIST 
10  '***** G.C.D  &  L.C.M *****GCD-2.BAS 
20  INPUT "INPUT A,B : ", A, B 
30  IF A < B THEN SWAP A,B 
40  BIG = A: SMALL = B 
50  R = BIG MOD SMALL 
60  IF R = 0 THEN 90 
70  BIG = SMALL : SMALL = R 
80  GOTO 50 
90  GCD = SMALL 
100 LCM = A * B / GCD 
110 PRINT "G.C.D. = "; GCD 
120 PRINT "L.C.M. = "; LCM 
Ok 

RUN 
INPUT A,B : 12,8 
G.C.D. =  4 
L.C.M. =  24 
Ok 
RUN 
INPUT A,B : 9,36 
G.C.D. =  9 
L.C.M. =  36 
Ok 

 
 
5. 氣泡排序法 (Bubble sort) 
 
【例題 6-11】 
 

LIST 
10 '***** BUBBLE SORT ***** BSORT-1.BAS 
20 CLS 
30 INPUT "How mary number : ", N 
40 DIM A(N) 
50 FOR I = 1 TO N 
60     PRINT I; " : ";: INPUT "", A(I) 
70 NEXT I 
80 ' 
90 '----------- SORT 
100 ' 
110 FOR I = N-1 TO 1 STEP -1 
120    FOR J = 1 TO I 
130        IF A(J) > A(J+1) THEN SWAP A(J), A(J+1) 
140    NEXT J 
150 NEXT I 
160 ' 
170 '---------- LIST DATA 
180 PRINT "***** SORT DATA *****" 
190 FOR I = 1 TO N 
200     PRINT I; ":"; A(I) 
210 NEXT I 
Ok 

RUN 
How mary number : 5 
 1  : 3 
 2  : 2 
 3  : 5 
 4  : 1 
 5  : 4 
***** SORT DATA ***** 
 1 : 1 
 2 : 2 
 3 : 3 
 4 : 4 
 5 : 5 
Ok 
 



  第六章  陣列 

BASIC 6 - 10 / 11 
Basic-6.doc曹維光 91年5月7日  

6. 插入排序法 (Insert sort) 
 
【例題 6-12】 
 

LIST 
10 '****** INSTER SORT ******** ISORT-1.BAS 
20 CLS 
30 INPUT "How mary number : ", N 
40 DIM A(N) 
50 FOR I = 1 TO N 
60     PRINT I; " : ";: INPUT "", A(I) 
70 NEXT I 
80 ' 
90 '----------- SORT 
100 ' 
110 FOR I = 1 TO N-1 
120    FOR J = I+1 TO N 
130        IF A(I) > A(J) THEN SWAP A(I), A(J) 
140    NEXT J 
150 NEXT I 
160 ' 
170 '---------- LIST DATA 
180 PRINT "***** SORT DATA *****" 
190 FOR I = 1 TO N 
200     PRINT I; ":"; A(I) 
210 NEXT I 
Ok 

RUN 
How mary number : 5 
 1  : 12 
 2  : 54 
 3  : 11 
 4  : 36 
 5  : 22 
***** SORT DATA ***** 
 1 : 11 
 2 : 12 
 3 : 22 
 4 : 36 
 5 : 54 
Ok 

【例題 6-13】 
 

LIST 
10 '***** TELSORT ***** TEL.BAS 
20 CLS 
30 N = 10 
40 DIM NAM$(N), TEL(N) 
50 FOR I = 1 TO N 
60     READ NAM$(I), TEL(I) 
70 NEXT I 
80 ' 
90 '---------------- DATA SORT 
100 ' 
110 FOR I = 1 TO N - 1 
120    FOR J = 1 TO I 
130        IF NAM$(I) < NAM$(J) THEN SWAP NAM$(I), NAM$(J): 
                               SWAP TEL(I), TEL(J) 
140    NEXT J 
150 NEXT I 
160 ' 
170 '--------------- DATA LIST 
180 ' 
190 PRINT "NO : NAME           TEL_NO " 
200 PRINT "== : ==========     =======" 
210 FOR I = 1 TO N 
220     PRINT USING "## : \        \     #######"; I; NAM$(I), TEL(I) 
230 NEXT I 
240 PRINT "== : ==========     =======" 
250 END 
260 ' 
300 DATA JOHN,3533591,NANCY,3022574 
310 DATA LUCK,7670609,MARY,3564784 
320 DATA ROBERT,3213414,BOB,9112437 
330 DATA ALICE,7210381,WANG,6532872 
340 DATA LEE,7676534,PAUL,5221050 
Ok 
RUN 
NO : NAME             TEL_NO 
== : ==========    ======= 
 1  : ALICE                 7210381 
 2  : BOB                    9112437 
 3  : JOHN                  3533591 
 4  : LEE                     7676534 
 5  : LUCK                 7670609 
 6  : MARY                3564784 
 7  : NANCY             3022574 
 8  : ROBERT            3213414 
 9  : WANG               6532872 
10 : PAUL                 5221050 
== : ==========    ======= 
Ok 

 


