
第五章  函數數值 
 

1. SQR (n)  {平方根} 
 
【例題 5-1】 

 
PRINT SQR(25) 
 5 
Ok 
PRINT SQR(2) 
 1.414214 
Ok 
 
PRINT 25^(1/2) 
 5 
Ok 
PRINT 8^(1/3) 
 2 
Ok 

 
 

2. ABS (變數)  {取絕對值} 
 
【例題 5-2】 

 
LIST 
10 A = -5 
20 PRINT A 
30 PRINT ABS(A) 
Ok 
RUN 
-5 
 5 
Ok 

【例題 5-3】畫磚紋 - 1 
 

LIST 
1  REM *** PRINT DRAW "*" -1 
****DRAW1.BAS 
10 FOR I = -5 TO 5 
20     Z = ABS(I) 
30     M = 11 - 2 * Z 
40     PRINT TAB(2+Z) 
50     FOR J = 1 TO M 
60         PRINT "*"; 
70     NEXT J 
80     PRINT 
90 NEXT I 
99 END 
Ok 

RUN 
      * 
     *** 
    ***** 
   ******* 
  ********* 
 *********** 
  ********* 
   ******* 
    ***** 
     *** 
      * 
Ok 

 
 
【例題 5-4】畫磚紋 - 2 
 

LIST 
1  REM *** PRINT DRAW "*" -2 
****DRAW2.BAS 
10 FOR I = -5 TO 5 
20     FOR J = -5 TO 5 
30         IF ABS(I) + ABS(J) <= 5 THEN PRINT "*"; 
                                   ELSE PRINT " "; 
40     NEXT J 
50     PRINT 
60 NEXT I 
99 END 
Ok 

RUN 

     * 

    *** 

   ***** 

  ******* 

 ********* 

*********** 

 ********* 

  ******* 

   ***** 

    *** 

     * 

Ok 

 
【例題 5-5】列出如下之磚紋 : 畫磚紋 - 3 
            * 
           * * 
          *   * 
         *     * 
        *       * 
       *         * 
        *       * 
         *     * 
          *   * 
           * * 
            * 



        第五章  函數數值 

BASIC 5 - 3 / 8 
Basic-5.doc曹維光 91年5月7日  

【解答 5-5】 
 

LIST 
1  REM *** PRINT DRAW "*" -3 
****DRAW3.BAS 
10 FOR I = -5 TO 5 
20     Z = ABS(I) 
30     M = 11 - 2 * Z 
40     PRINT TAB(2+Z) 
50     FOR J = 1 TO M 
60         IF J = 1 OR  J = M THEN  PRINT "*"; 
                              ELSE  PRINT " "; 
70     NEXT J 
80     PRINT 
90 NEXT I 
99 END 
Ok 

RUN 
      * 
     * * 
    *   * 
   *     * 
  *       * 
 *         * 
  *       * 
   *     * 
    *   * 
     * * 
      * 
Ok 

 
 
【例題 5-6】畫磚紋 - 4 
 

1  REM *** PRINT DRAW "*" -4 
****DRAW4.BAS 
10 FOR I = -5 TO 5 
20     FOR J = -5 TO 5 
30         IF ABS(I) + ABS(J) = 5 THEN PRINT "*"; 
                                  ELSE PRINT " "; 
40     NEXT J 
50     PRINT 
60 NEXT I 
99 END 
Ok 

RUN 
     * 
    * * 
   *   * 
  *     * 
 *       * 
*         * 
 *       * 
  *     * 
   *   * 
    * * 
     * 
Ok 

3. INT(n)  {取整數} 
 
4. FIX(n)  {去小數} 
 
【例題 5-7】 
 

INT FIX 
PRINT INT(5.6) 
5 
Ok 
PRINT INT(-5.6) 
-6  
Ok 

PRINT FIX(5.6) 
5 
Ok 
PRINT FIX(-5.6) 
-5 
Ok 

 
 
【例題 5-8】輸入一數值 N, 判斷是否為整數   
 

LIST 
10 '*** INT test *** int.bas 
20 INPUT "ENTER A NUMBER : ", N 
30 IF N = INT(N) THEN PRINT "INTEGER !!!" 
                 ELSE PRINT "NOT INTEGER !!!" 
40 END 
Ok 
RUN 
ENTER A NUMBER : 12 
INTEGER !!! 
Ok 
RUN 
ENTER A NUMBER : 5 
INTEGER !!! 
Ok 
RUN 
ENTER A NUMBER : 66.4 
NOT INTEGER !!! 
Ok 

 
5. READ 變數 
       : 
       : 
       : 
    DATA 資料1,資料2,....資料n 
 
        變數 ==> 可為數值或字串變數 
        資料 ==> 可為數值或字串常數 

【例題 5-9】 
 

LIST 
10 '*** READ ... DATA .. **** read.bas 
20 FOR I = 1 TO 5 
30     READ A 
40     PRINT A; 
50 NEXT I 
99 END 
100 ' 
110 DATA 1,2,5,6,3 
Ok 
RUN 
 1  2  5  6  3 
Ok 

 
 
【例題 5-10】 
 

LIST 
10 '*** READ ... DATA .. **** read-1.bas 
20 FOR I = 1 TO 5 
30     READ A, NAM$ 
40     PRINT A, NAM$ 
50 NEXT I 
99 END 
100 ' 
110 DATA 1001,"JOHN" 
120 DATA 1002,"MARY" 
130 DATA 1003,"WANG" 
140 DATA 1004,"LEE" 
150 DATA 1005,"BOB" 
Ok 
RUN 
 1001         JOHN 
 1002         MARY 
 1003         WANG 
 1004         LEE 
 1005         BOB 
Ok 

【例題 5-11】 
 

LIST 
10 '************ STU-2 ********** STU-2.BAS 
20 CLS 
30 PRINT "INPUT DATA : " 
40 FOR I = 1 TO 20 
50     READ S 
60     PRINT S, 
70 IF S >= 90 THEN S9 = S9 + 1 
      ELSE IF S >= 80 THEN S8= S8 + 1 
        ELSE IF S >= 70 THEN S7 = S7 + 1 
          ELSE IF S >= 60 THEN S6 = S6 + 1 
            ELSE S5 = S5 + 1 
80 NEXT I 
90 PRINT 
100 ' 
110 PRINT "100 - 90 : "; S9 
120 PRINT " 89 - 80 : "; S8 
130 PRINT " 79 - 70 : "; S7 
140 PRINT " 69 - 60 : "; S6 
150 PRINT " 59 -    : "; S5 
160 ' 
200 DATA 56,67,83,92,43 
210 DATA 82,94,77,78,63 
220 DATA 69,76,58,65,72 
230 DATA 81,64,73,61,56 
Ok 
RUN 
INPUT DATA : 
 56            67            83            92            43 
 82            94            77            78            63 
 69            76            58            65            72 
 81            64            73            61            56 
 
100 - 90 :  2 
  89 - 80 :  3 
  79 - 70 :  5 
  69 - 60 :  6 
  59 -      :  4 
Ok 

【例題 5-12】 
 

LIST 
10 '********** STU-4 ********* STU-4.BAS 
20 PRINT "  NO "; TAB(9); "S1"; TAB(13); "S2"; TAB(17); "S3"; 
         TAB(21); "S4"; TAB(26); "TOTAL"; TAB(33); " AVG " 
30 PRINT " ===="; TAB(9); "=="; TAB(13); "=="; TAB(17); "=="; 
         TAB(21); "=="; TAB(25); "====="; TAB(32); "=====" 
40 FOR I = 1 TO 5 
50     READ NO, S1, S2, S3, S4 
60     SUM = S1 + S2 + S3 + S4 
70     AVG = SUM / 4 
80     PRINT NO; TAB(8); S1; TAB(12); S2; TAB(16); S3; 
             TAB(20); S4; TAB(24); SUM; TAB(31); AVG 
90 NEXT I 
100 PRINT " ===="; TAB(9); "=="; TAB(13); "=="; TAB(17); "=="; 
         TAB(21); "=="; TAB(25); "====="; TAB(32); "=====" 
110 DATA 1001,62,70,89,75 
120 DATA 1002,69,82,74,63 
130 DATA 1003,72,93,61,65 
140 DATA 1004,75,66,84,94 
150 DATA 1005,67,89,56,72 
Ok 
RUN 
  NO    S1   S2  S3  S4  TOTAL  AVG 
 ====  ==  ==  ==  ==  =====  ===== 
 1001   62   70  89   75     296    74 
 1002   69   82  74   63     288    72 
 1003   72   93  61   65     291    72.75 
 1004   75   66  84   94     319    79.75 
 1005   67   89  56   72     284    71 
 ====  ==  ==  ==  ==  =====  ===== 
Ok 

6. RESTORE [line no.]  {重覆讀取資料} 
 
    • 須配合 READ ... DATA ... 使用 
    • 不加行號時, 再從第一個 DATA 讀取 
    • 加行號時, 從指定行號的 DATA 讀取 
 
【例題 5-13】 
 

LIST 
10 '*** restore test *** restore1.bas 
20 READ A,B 
30 PRINT A,B 
40 READ C 
50 PRINT C 
60 READ D,E 
70 PRINT D,E 
80 DATA 1,2 
90 DATA 3 
Ok 
RUN 
 1             2 
 3 
Out of DATA in 60 

 
 
【例題 5-14】 
 

LIST 
10 '*** restore test *** restore1.bas 
20 READ A,B 
30 PRINT A,B 
40 READ C 
50 PRINT C 
55 RESTORE 
60 READ D,E 
70 PRINT D,E 
80 DATA 1,2 
90 DATA 3 
Ok 
RUN 
 1             2 
 3 
 1             2 
Ok 

 


