
第九章  其他函數 
 
1. ON n GOTO 行號1, 行號2,......行號m 
 
     n = 1 時, GOTO 至行號1 
     n = 2 時, GOTO 至行號2 
       :         : 
     n = m 時, GOTO 至行號m 
 
【例題 9-1】 
 

LIST 
10 '*** ON GOTO TEST *** ongoto.bas 
20 CLS: KEY OFF 
30 LOCATE  3,32: PRINT "MAIN MENU" 
40 LOCATE  4,32: PRINT "=========" 
50 LOCATE  6,25: PRINT "1: A + B" 
60 LOCATE  6,40: PRINT "2: A - B" 
70 LOCATE  8,25: PRINT "3: A * B" 
80 LOCATE  8,40: PRINT "4: A / B" 
90 LOCATE 10,25: PRINT "0: EXIT"; 
100 LOCATE 12,32: PRINT "CHOICE : "; 
110 ' 
120 K$ = INPUT$(1): PRINT K$ 
130 IF INSTR("01234",K$)= 0 THEN BEEP: GOTO 100 
140 IF K$ = "0" THEN 230 
150 ' 
160 LOCATE 14,32: INPUT "ENTER A,B : ", A,B 
170 COLOR 0,7 
180 LOCATE 18,32: PRINT "PRESS ANY KEY ..." 
190 COLOR 7,0 
200 LOCATE 16,32: 
210 N = VAL(K$) 
220 ON N GOTO 1000, 2000, 3000, 4000 
230 END 
1000 '-------- A + B 
1010 PRINT A; "+"; B; "="; A + B 
1020 T$ = INPUT$(1) 
1030 GOTO 20 
2000 '-------- A - B 
2010 PRINT A; "-"; B; "="; A - B 
2020 T$ = INPUT$(1) 
2030 GOTO 20 
3000 '-------- A * B 
3010 PRINT A; "*"; B; "="; A * B 
3020 T$ = INPUT$(1) 
3030 GOTO 20 
4000 '-------- A / B 
4010 PRINT A; "/"; B; "="; A / B 
4020 T$ = INPUT$(1) 
4030 GOTO 20 
Ok 

RUN 
       MAIN MENU 
       ========= 
 
1: A + B       2: A - B 
 
3: A * B       4: A / B 
 
0: EXIT 
 
       CHOICE : 1 
 
       ENTER A,B : 12,34 
 
        12 + 34 = 46 
 

       PRESS ANY KEY ... 

  第九章  其他函數 

2. ON n GOSUB 行號1, 行號2,......行號m 
 
     n = 1 時, GOSUB 至行號1 
     n = 2 時, GOSUB 至行號2 
       :         : 
     n = m 時, GOSUB 至行號m 
 
【例題 9-2】 

 
LIST 
10 '*** ON GOSUB TEST *** ongosub.bas 
20 CLS: KEY OFF 
30 LOCATE  3,32: PRINT "MAIN MENU" 
40 LOCATE  4,32: PRINT "=========" 
50 LOCATE  6,25: PRINT "1: A + B" 
60 LOCATE  6,40: PRINT "2: A - B" 
70 LOCATE  8,25: PRINT "3: A * B" 
80 LOCATE  8,40: PRINT "4: A / B" 
90 LOCATE 10,25: PRINT "0: EXIT"; 
100 LOCATE 12,32: PRINT "CHOICE : "; 
110 ' 
120 K$ = INPUT$(1): PRINT K$ 
130 IF INSTR("01234",K$)= 0 THEN BEEP: GOTO 100 
140 IF K$ = "0" THEN 230 
150 ' 
160 LOCATE 14,32: INPUT "ENTER A,B : ", A,B 
170 COLOR 0,7 
180 LOCATE 18,32: PRINT "PRESS ANY KEY ..." 
190 COLOR 7,0 
200 LOCATE 16,32: 
210 N = VAL(K$) 
220 ON N GOSUB 1000, 2000, 3000, 4000 
225 GOTO 20 
230 END 
1000 '-------- A + B 
1010 PRINT A; "+"; B; "="; A + B 
1020 T$ = INPUT$(1) 
1030 RETURN 
2000 '-------- A - B 
2010 PRINT A; "-"; B; "="; A - B 
2020 T$ = INPUT$(1) 
2030 RETURN 
3000 '-------- A * B 
3010 PRINT A; "*"; B; "="; A * B 
3020 T$ = INPUT$(1) 
3030 RETURN 
4000 '-------- A / B 
4010 PRINT A; "/"; B; "="; A / B 
4020 T$ = INPUT$(1) 
4030 RETURN 
Ok 

RUN 
       MAIN MENU 
       ========= 
 
1: A + B       2: A - B 
 
3: A * B       4: A / B 
 
0: EXIT 
 
       CHOICE : 3 
 
       ENTER A,B : 2,3 
 
        2 * 3 = 6 
 

       PRESS ANY KEY ... 

  第九章  其他函數 

【例題 9-3】賽車遊戲 
 

LIST 
10 '*** CAR GAME *** cargame.bas 
20 CLS 
30 KEY OFF: CLS 
40 R$ = CHR$(176): C$ = ":H:" 
50 LP = 35: RP = 45: W = 10 
60 X = 40: Y = 10: P = 30: Q = 50 
70 SPEED = 150 
80 DIM A(15), B(15) 
90 ' 
100 ' 
110 FOR I = 1 TO 15 
120     A(I) = LP: B(I) = RP 
130 NEXT I 
140 LOCATE 25, I: PRINT 
150 ' 
160 ' 
170 K$ = INKEY$: SC = SC + 1 
180 LOCATE Y-1, X: PRINT "   " 
190 IF K$ = "," THEN X = X - 1: SOUND 80 * X, .5 
200 IF K$ = "." THEN X = X + 1: SOUND 80 * X, .5 
210 LOCATE Y, X: PRINT C$: SOUND 1000, .03 
220 RANDOMIZE VAL(RIGHT$(TIME$, 2)) ^ 2 
230 R1 = INT (RND * 3) + 1 
240 R2 = INT (RND * 3) + 1 
250 T = T + 1 
260 LP = LP + R1 - 2 
270 RP = RP + R2 - 2 
280 IF LP < P THEN LP = P 
290 IF RP > Q THEN RP = Q 
300 IF RP - LP < W THEN LP = LP + 2 - R1: RP = RP + 2 - R2 
310 IF T > 15 THEN T = 1 
320 A(T) = LP: B(T) = RP 
330 IF T = 15 THEN S = 1 ELSE S = T + 1 
340 IF SC MOD 100 = 0 THEN CO = SC MOD 3 
350 R$ = CHR$(176+CO) 
360 LOCATE 24, 1: PRINT TAB(LP); R$; TAB(RP); R$ 
370 IF X <= A(S) OR X + 2 >= B(S) THEN LOCATE Y, X: PRINT "@@@": 
                  PRINT TAB(X-4); "SCORE : "; SC: END 
380 FOR I = 1 TO SPEED: NEXT I 
390 IF SPEED > 1 THEN SPEED = SPEED - 1 
400 IF SC MOD 30 = 0 THEN W = W - 1: 
                    IF W < 5 THEN W = 5 
410 GOTO 170 
Ok 

